Electronic Supplementary Material (ESI) for Analyst.

Supplementary information for
Spatially encoded diffusion-ordered NMR spectroscopy of reaction mixtures in organic solvents

Ghanem Hamdoun,[a] Ludmilla Guduff,[a] Carine van Heijenoort,[a] Christophe Bour,[b]
Vincent Gandon,[b] Jean-Nicolas Dumez[a]

[a] *Institut de Chimie des Substances Naturelles. CNRS UPR2301. Univ. Paris Sud. Université Paris-Saclay. 91190 Gif-Sur-Yvette France.*
[b] *Institut de Chimie Moléculaire et des Matériaux d'Orsay. CNRS UMR 8182. Univ. Paris Sud. Université Paris-Saclay. 91405 Orsay France.*

I- Supplementary figures



Figure S1 Signal decay as a function of $\Delta'(K(z))^2$, where Δ' is the effective diffusion delay and *K*(*z*) is the effective gradient area (see Ref. [1]), for the cyclododecane resonance for the SPEN STE DOSY (blue) and SPEN DSTE DOSY (red) experiments carried out in CD_2Cl_2. The diffusion delay was 140 ms for both SPEN STE and DSTE experiment.



Figure S2. Signal attenuation as a function of the squared diffusion gradient amplitude for selected resonances of the mono- and di-imine compounds, obtained with a conventional DOSY sequence, in CD_3CN, during the course of a di-amination reaction.



Figure S3 DOSY spectra of the reaction mixture in CD_3CN at 283 K, at equilibrium, obtained with (a) the conventional DSTE DOSY pulse sequence using stoichiometry i/; (b) the conventional DSTE DOSY experiment using stoichiometry ii/ (see sec. B for the stoichiometry).

II- Pulse sequence for spectrally selective SPEN DOSY (ssSPEN DSTE)

```
;ssSPENDSTE
#include <Avance.incl>
#include <Grad.incl>
#include <De.incl>
#include <Delay.incl>

"d20=(td*dw/(2*l3))-d6"
"DELTA1=d30/2-p11-2*d4-d11-2*p20-3*d16-p22-2*p1"
"p2=p1*2"

1 ze
  20u st0
  30m zd
2 30m
  d1
  50u UNBLKGRAD
  20u pl1:f1
;----------------------------------------------excitation
  p1 ph1
  d5
;----------------------------------------------first spin echo
;----------------------------------------------spatial encoding
  p20:gp20
  d16 pl0:f1
  d4 gron0
  p11:sp1:f1 ph2
  d11
  d4 groff
  p20:gp21
  d16 pl1:f1
  p1 ph1
  p22:gp14
  d16
  DELTA1
  p1 ph1
;----------------------------------------------decoding period
  p20:gp22
  d16 pl0:f1
  d4 gron0
  p11:sp1:f1 ph2
  d11
  d4 groff
  p20:gp23
  d16 pl1:f1
;----------------------------------------------Middle 180 pulse
  p2 ph3
;----------------------------------------------second spin echo
;----------------------------------------------spatial encoding
  p20:gp26
  d16 pl0:f1
  d4 gron0
  p11:sp1:f1 ph2
  d11
  d4 groff
  p20:gp27
  d16 pl1:f1
  p1 ph1
  p22:gp15
  d16
  DELTA1
  p1 ph1
;----------------------------------------------decoding period
  p20:gp28
  d16 pl0:f1
  d4 gron0
  p11:sp1:f1 ph2
  d11
```

```
  d4 groff
  p20:gp29
  d16
;---------------------------------------------region selection
  p16:gp30
  d16 pl1:f1
  p12:sp2:f1 ph4
  p16:gp31
  d16
;---------------------------------------------pre-phasing
  p25:gp25
  d16
;---------------------------------------------acquisition
   ACQ_START(ph30,ph31)
   1u DWELL_GEN:f1
3 d20 gron2
   d6 groff
   d20 gron3
   d6 groff
   lo to 3 times l3
   rcyc=2
   30m mc #0 to 2 F1QF(id2)
   100u BLKGRAD
   d17

;---------------------------------------------phase cycling
ph1=0
ph2=0
ph3=0
ph4=0
ph29=0
ph30=0
ph31=0

;pl1 : f1 channel - power level for pulse (default)
;sp1: shaped pulse power level for selective detection
;p1 : f1 channel - 90 degree high power pulse
;p11: duration of the encoding chirp
;p20 :1ms coherence-selection delay
;p22 :20ms coherence-selection delay
;spnam1 : shaped pulse for spatial encoding
;d1: relaxation delay; 1-5 * T1
;d2: ramping period for gradient pulse (25us)
;d3: delay 10us
;d4: delay 25us
;d5: delay 5us
;d17: delay 10us
;d16: gradient recovery delay (200us)
;d30: diffusion time (big DELTA)
;d6: gradient ramp off during acauisition (25us)
;d20 + d6 : acquisition gradient duration
;GPZ0 : strength for excitation gradient
;GPZ2 and GPZ3 : strength for reversed acquisition gradient GPZ3 = -GPZ2
;GP20 and GP26 : coherence-selection gradients encoding (a)
;GP21 and GP27 : coherence-selection gradients encoding (a+c)
;GP22 and GP28 : coherence-selection gradients encoding (b)
;GP23 and GP29 : coherence-selection gradients encoding (b+c)
;GP30 and GP31 : coherence-selection gradients GP30 = GP31
;GP14 : spoiler Gradient Pulse
;GPZ15 : pre-phasing gradient GPZ15= -GPZ2/2
;GPNAM14= GPNAM15= GPNAM20= GPNAM21= GPNAM22= GPNAM23= GPNAM26= GPNAM27= SINE.100
;GPNAM28= GPNAM30= GPNAM31= SINE.100
;GPNAM25= RECT.1
;l3=number of loops for acquisition
;IMPORTANT: set d20 + d6 = DW x TD(f3)/(2xL3)
```

1. L. Guduff, I. Kuprov, C. van Heijenoort and J.-N. Dumez, *Chem. Commun.*, 2017, **53**, 701-704.